

BAB 3

Melakukan Perhitungan dengan Formula dan Fungsi

Mengenal Formula

Formula atau rumus, adalah alat yang memungkinkan Anda untuk melakukan kalkulasi terhadap nilai-nilai pada sel itu sendiri (sel yang ditempati oleh formula itu) maupun nilai yang tersimpan pada sel-sel yang lain.

Bagaimana cara kerja formula. Sebuah formula dapat membantu Anda menganalisis data pada worksheet. Anda dapat melakukan perhitungan seperti penambahan, perkalian dan bahkan perbandingan nilai-nilai yang terdapat pada worksheet. Anda dapat menggunakan formula, apabila perlu memasukkan nilai kalkulasi pada worksheet Anda. Sebuah formula dapat terdiri atas elemen-elemen berikut ini:

- operator perhitungan
- referensi alamat suatu sel
- nilai dari angka-angka secara langsung
- fungsi Excel
- dan juga nama sel/nama range

Untuk memasukkan suatu formula pada suatu sel, ketikkan kombinasi dari elemen-elemen di atas, dan ketikkan itu akan muncul langsung pada sel yang ditempati dan juga pada formula bar. Sebuah formula harus selalu diawali dengan karakter = (sama dengan).

Mengenal Empat Kelompok Formula

Excel membagi formula ke dalam empat kelompok, yaitu Aritmetik, Perbandingan, Teks, dan Referensi. Masing-masing kelompok memiliki perangkat operator yang memiliki cara kerja berbeda.

Menggunakan Formula Aritmetik

Formula aritmetik adalah formula yang paling umum digunakan. Formula ini menggabungkan, baik komponen berupa bilangan, alamat sel, dan juga fungsi Excel, kemudian melakukan operasi dengan salah satu operator matematika sehingga akan menghasilkan hasil kalkulasi.

Berikut ini operator yang dipergunakan dalam formula aritmetik.

Operator	Nama	Contoh	Hasil
+	Penjumlahan	=25+5	30
-	Pengurangan	=30-10	10
-	Nilai Negatif	=-75	-75
*	Perkalian	=5*25	125
/	Pembagian	=50/10	5
%	Persentase	=125%	0.125
^	Perpangkatan	=5^5	3125

Sebagian besar cara kerja operator aritmetik di atas sudah umum, namun cara kerja operator perpangkatan yang sedikit berbeda. Misalnya $=x^y$ berarti nilai x dikalikan sebanyak y kali. Contoh formula $=3^2$ menghasilkan $3*3$, yaitu 9. Sama halnya dengan $=2^4$ akan menghasilkan $2*2*2*2= 16$.

Menggunakan Formula Perbandingan

Formula perbandingan adalah suatu pernyataan yang membandingkan dua atau beberapa bilangan, teks maupun nilai yang terkandung pada sel, maupun hasil perhitungan suatu fungsi. Jika statement (pernyataan) menghasilkan nilai logika TRUE maka formula akan memberikan nilai “bukan nol”, artinya akan menampilkan hasil perhitungan dalam bentuk nilai.

Sebaliknya, jika nilai logika menghasilkan FALSE, maka akan ditampilkan sebagai nilai nol.

Berikut tabel yang memuat operator-operator perbandingan yang digunakan dalam kelompok Formula Perbandingan.

Operator	Nama	Contoh	Hasil
=	Sama dengan	=10=5	FALSE
>	Lebih besar dari	=10>5	TRUE
<	Lebih kecil dari	=-10<5	FALSE
>=	Lebih besar sama dengan	=”a”>=”b”	FALSE
<=	Lebih kecil sama dengan	=”a”<=”b”	TRUE
<>	Tidak sama dengan	=”a”<>”b”	TRUE

Formula perbandingan memiliki banyak kegunaan. Sebagai contoh, Anda dapat menentukan apakah membayar bonus wiraniaga dengan menggunakan formula perbandingan untuk membandingkan actual sales dengan target yang ditentukan. Jika wiraniaga menjual lebih

banyak dari target yang ditentukan, maka ia akan mendapatkan bonus. Anda juga dapat memonitor tagihan kredit, misalnya jika sejumlah pelanggan telah melampaui tanggal jatuh tempo lebih dari 150 hari, maka Anda akan mengirimkan tagihan ke bagian penagihan agar dilakukan penagihan.

Menggunakan Formula Teks

Berbeda dengan dua jenis formula yang disampaikan di depan, yang dalam kalkulasi menghasilkan nilai, maka formula teks akan menghasilkan teks juga. Formula teks menggunakan operator “&” untuk bekerja dengan sel yang berisi teks, karakter lain termasuk di dalamnya tanda petik dan hasil dari fungsi teks.

Formula teks gunanya untuk merangkai antara data teks dengan data teks lainnya. Data bisa berupa data yang diketik langsung pada formula, maupun teks yang tersimpan dalam suatu sel.

Contoh:

=”Microsoft “&” Excel” menghasilkan Microsoft Excel

=”Microsoft”&”Excel” menghasilkan MicrosoftExcel

Perbedaan saat menulis formula teks yang pertama pada kata Microsoft memberikan spasi sebelum tanda petik, dan memberikan spasi setelah tanda petik dan sebelum kata Excel. Hasilnya antara Microsoft dan Excel terdapat dua spasi kosong. Sedangkan pada contoh formula yang kedua, tidak ada spasi di antara kedua kata.

Anda juga bisa memberikan spasi secara khusus di antara dua data dengan menuliskan “ “.

Contoh:

=”Microsoft”&” “&”Excel” menghasilkan Microsoft Excel.

Jika A1 berisi Microsoft dan B1 berisi Excel, maka =A1&” “&B1 menghasilkan Microsoft Excel.

Beberapa Contoh Mengaplikasikan Formula

Berikut ini beberapa contoh bagaimana mengaplikasikan formula (saja) tanpa menggunakan fungsi.

1. Menghitung Harga Discount

Semua barang yang tertera pada Daftar Harga di-discount sebesar 15%, di mana besarnya discount ditulis pada sel C3.

Langkah pengerjaannya sebagai berikut:

1. Buatlah tabel seperti Gambar 3.1 berikut ini.
2. Besarnya discount pada sel C3 ditulis **-15%** karena dipergunakan untuk mengurangi harga satuan.
3. Bloklah sel C5:C10 lalu pada Formula Bar ketikkan formula **=B5+(B5*\$C\$3)**.
4. Tekan **<Ctrl+Enter>**.

	A	B	C
1	Daftar Harga		
2			
3		Discount	-15%
4	Kode barang	Harga Satuan	Harga setelah Discount
5	AG-115	75000	63750
6	AG-116	50000	42500
7	AF-207	95000	80750
8	AD-100	45000	38250
9	AG-120	80000	68000
10	AD-103	50000	42500
11			

Gambar 3-1: Tabel untuk menghitung harga setelah di-discount.


2. Mengonversi Nilai Mata Uang US\$ ke Rp

Nilai tukar dengan mata uang asing selalu berubah setiap saat. Untuk itu, daftar harga yang berasal dari mata uang US\$ perlu dikonversikan ke mata uang Rupiah.

Pada worksheet Gambar 3.2 berikut ini, nilai tukar yang berlaku saat itu diletakkan pada sel C3.

Untuk membuat nilai konversi pada worksheet lakukan langkah-langkah sebagai berikut:

1. Susunlah tabel seperti gambar berikut.
2. Pilihlah sel C6:C20.
3. Pada formula bar ketikkan formula: **=B6*\$C\$3** lalu tekan **<Ctrl+Enter>**.



The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1	KONVERSI NILAI DOLLAR KE RUPIAH		
3	Nilai Tukar US\$ ke Rupiah hari ini		9200
5	Judul Buku	Harga US\$	Harga Rp
6	Brilliant Digital Photography For The Over 50s	40.54	372,968
7	Brilliant Digital Photography	40.54	372,968
8	Nash Editions: Photography And The Art Of Digital	54.35	500,020
9	The Digital Photography Book: Step-By-Step Secrets	29.89	274,988
10	The Digital Photography Book, Vol 2: Step-By-Step	29.89	274,988
11	The Moment It Clicks: Photography Secrets From	65.00	598,000
12	The Hot Shoe Diaries: Big Light From Small Flashes	47.83	440,036
13	Scott Kelby's Digital Photography Boxed Set: The	54.13	497,996
14	Within The Frame: The Journey Of Photographic	47.83	440,036
15	Canon 50D From Snapshots To Great Shots	32.34	297,528
16	Nikon D200 Digital Field Guide	25.00	230,000
17	Exposure And Lighting For Digital Photographers	43.26	397,992
18	Nikon Creative Lighting System Digital Field Guide	25.33	233,036
19	Canon Speedlite System Digital Field Guide	25.00	230,000
20	Canon EOS 30D Digital Field Guide	25.33	233,036

Gambar 3-2: Konversi dari harga dalam nilai US\$ ke nilai Rupiah.

Catatan:

Untuk membuat konversi sebaliknya, dari nilai Rupiah ke US\$ gunakan formula =C6/\$C\$3.

3. Menghitung Konsumsi BBM

Suatu analisis yang sering dilakukan untuk mengetahui seberapa efektif pemakaian BBM dari suatu kendaraan adalah mengetahui berapa jarak tempuh yang diperoleh dari konsumsi BBM setiap liter pada kendaraan. Berikut ini contoh langkah-langkah pembuatan worksheet untuk melakukan perhitungan tersebut.

1. Buatlah worksheet seperti Gambar 3.3 berikut ini.
2. Bloklah sel D4:D9, lalu pada Formula Bar ketikkan formula berikut =B4/C4.
3. Tekan <Ctrl+Enter>.

	A	B	C	D
1	Jarak Tempuh yang diperoleh BBM setiap Liter			
2				
3	Tanggal	Jarak yang ditempuh dlm KM	Konsumsi BBM dalam Liter	Rata-rata jarak tempuh per liter
4	11-Mar	813	114	7.13
5	12-Mar	901	115	7.83
6	13-Mar	821	110	7.46
7	15-Mar	966	117	8.26
8	17-Mar	885	106	8.35
9	18-Mar	803	114	7.04

Gambar 3-3: Menghitung berapa jarak tempuh dalam KM setiap 1 liter BBM.


Mengenal Fungsi Excel

Fungsi Excel atau kita sebut Fungsi adalah formula atau rumus siap pakai untuk menyelesaikan suatu tugas perhitungan tertentu. Fungsi dibuat untuk menyederhanakan perhitungan yang panjang dan kadang-kadang rumit. Sebagai contoh:

1. Formula yang dibuat sendiri: $=A1+A2+A3+A4+A5$ dapat ditulis menggunakan fungsi $=SUM(A1:A5)$.
2. Formula yang dibuat sendiri: $=A1*A2*A3*A4*A5$ dapat dihitung menggunakan fungsi $=PRODUCT(A1:A5)$.
3. Formula yang dibuat sendiri $=(A1+A2+A3+A4+A5)/5$ dapat ditulis menggunakan fungsi $=AVERAGE(A1:A5)$.

Aturan Penulisan Fungsi dalam Formula


1. Apabila suatu fungsi terletak di depan suatu formula, maka harus diawali dengan simbol = (sama dengan) sebagai syarat awal penulisan formula.
2. Tanda kurung buka dan kurung tutup (parentheses) untuk memberitahukan Excel, di mana argumen dimulai dan diakhiri. Penulisan kurung buka dan kurung tutup tanpa diawali dan diakhiri dengan spasi.



3. Setiap argumen dari suatu fungsi harus ditulis di antara dua tanda kurung. Argumen dapat terdiri atas bilangan, teks, nilai logika,

array maupun referensi dan nama sel/range. Keterangan masing-masing argumen pada buku ini kami tulis di bawah masing-masing Bentuk Penulisan. Argumen selain berupa unsur-unsur di atas, juga dapat berupa hasil suatu fungsi yang lain. Penggunaan fungsi secara berganda ini juga disebut sebagai nested function. Berikut ini contohnya. Microsoft Excel dapat memuat hingga tujuh buah fungsi berganda dalam sebuah formula.

4. Susunan sebuah formula dapat terdiri atas paling banyak 1024 karakter (Excel 2007) dan 8,192 (Excel 2010) termasuk tanda =, nama fungsi, tanda kurung dan seluruh argumen maupun pemisah argumen dan operator-operatornya. Jika Anda perlu menuliskan karakter lebih dari batasan tersebut, maka pecahlah menjadi beberapa formula yang terletak dalam sel-sel yang berbeda, kemudian jadikan referensi pada formula lain yang Anda buat. Sebuah formula panjang dalam satu sel apabila dijadikan referensi bagi formula lain, cukup ditulis dengan menyebutkan alamat sel di mana formula itu berada.



5. Excel melakukan penulisan nama formula menggunakan huruf kapital. Apabila Anda menggunakan huruf kecil, maka setelah menekan Enter, secara otomatis Excel akan mengubahnya menjadi huruf kapital. Apabila ternyata ketikan huruf kecil itu tidak dikonversi menjadi kapital, artinya Excel tidak menerima fungsi yang Anda ketikkan karena dianggap ada kesalahan ketik. Untuk itu periksa kembali pengetikan nama fungsinya.
6. Jika Excel tidak dapat menerima formula yang Anda ketikkan dan menampilkan nilai kesalahan, maka kemungkinannya adalah:

- Ada tanda kurung yang tidak seimbang.
- Ada argumen yang tidak tertulis, atau argumen kurang lengkap.
- Penulisan argumen tidak sebagaimana yang disyaratkan oleh fungsi, misal data yang seharusnya berupa bilangan tetapi tertulis sebagai teks dan kesalahan yang sejenis.

Macam-Macam Fungsi pada Excel

Hingga Excel 2010 telah memiliki ratusan buah fungsi yang dikelompokkan ke dalam masing-masing lingkup kerjanya sebagai berikut:

1. **Fungsi Financial**, menampilkan fungsi-fungsi Excel untuk melakukan perhitungan di bidang keuangan.
2. **Fungsi Date & Time**, menampilkan fungsi-fungsi Excel untuk melakukan perhitungan tanggal dan waktu.
3. **Fungsi Math & Trig**, untuk menampilkan pilihan fungsi-fungsi Excel di bidang Matematika dan Trigonometri.
4. **Fungsi Statistical**, fungsi-fungsi Excel di bidang Statistika.
5. **Fungsi Lookup & Reference**, menampilkan fungsi-fungsi Excel yang berfungsi untuk menampilkan informasi berdasarkan kriteria tertentu dari suatu table.
6. **Database & Table**, untuk melakukan perhitungan dalam suatu lingkup tabel data.
7. **Fungsi Text**, untuk menampilkan fungsi-fungsi Excel untuk mengelola dan mencari teks yang tersimpan dalam suatu sel.
8. **Fungsi Logical**, untuk menampilkan fungsi-fungsi Excel dalam bidang Logika, yaitu untuk melakukan perbandingan dan hubungan logis antara dua kondisi.

9. **Fungsi Information**, fungsi-fungsi Excel untuk menghasilkan informasi tentang sel, range, baik dalam hal letak maupun format yang berlaku.
10. **Fungsi Engineering**, fungsi-fungsi Excel di bidang Engineering.
11. **Fungsi Cube**, fungsi-fungsi Excel untuk menghitung volume berpangkat tiga. Ini adalah kelompok fungsi baru yang ada pada Excel 2010.
12. **Compatibility**, (pada Excel 2010), terdiri atas kumpulan fungsi lama yang berlaku pada Excel 2007 ke belakang. Sementara pada Excel 2010, fungsi-fungsi ini telah disempurnakan dan menggunakan nama baru. Pada Excel 2010, fungsi-fungsi ini masih tetap dapat dipergunakan.

Contoh Penggunaan Fungsi Tanggal dan Waktu

1. Menghitung Jam dengan Bantuan Fungsi NOW

Apabila Anda bermaksud mengalkulasi waktu dimulai saat sekarang, maka fungsi NOW akan sangat berguna. Perlu kita ingat bahwa Excel menyimpan waktu sebagai nilai desimal. Sebagai contoh: 0.5 adalah sama dengan jam 12:00 siang hari, 0.75 sama dengan jam 18:00, demikian seterusnya. Selain pemahaman tersebut, 0.5 berarti sama dengan nilai 12 jam, sedang 0.75 sama dengan nilai 18 jam.

Berikut ini contoh pengaplikasiannya pada worksheet.

1. Pada sel C3 isikan formula: **=NOW()**
2. Pada sel C4 isikan formula: **=C3+0.25** (0.25 nilainya sama dengan 6 jam)
3. Pada sel D5 isikan formula: **=C3+1** (1 nilainya sama dengan 24 jam)

	A	B	C	D	E
1	MENGHITUNG WAKTU DENGAN FUNGSI NOW				
2					
3	Mulai Lomba		3/18/2010 10:48		
4	Selesai Lomba		3/18/2010 16:48		
5	Lomba sesi berikutnya dimulai jam			3/19/2010 10:48	
6					

Gambar 3-4: Menghitung waktu dengan fungsi NOW.

2. Menggunakan Fungsi DATEVALUE untuk Mengonversi Tanggal yang Ditulis dalam Format Teks

Suatu hal yang perlu kita ketahui bahwa Excel semakin berkembang dalam menerima masukan data tanggal dan waktu sehingga mudah diterima dalam format yang dimengerti Excel. Pada Excel versi-versi terdahulu, sebuah data tanggal yang diimpor dari program database maupun pengolah kata tidak serta-merta dapat diterima sebagai data dalam format tanggal sehingga perlu dikonversi dengan fungsi DATEVALUE. Penulisan data tanggal dalam format teks kemungkinannya bisa bermacam-macam. Berikut ini salah satu kasus untuk latihan dengan langkah-langkah sebagai berikut:

1. Ketikkan data-data tanggal di kolom A dan B dengan terlebih dulu membubuhkan karakter ' di depannya. Jadi, '11/03/03 dan '12/04/2003. Gunanya agar data yang kita ketikkan akan berformat Teks dan bukan data tanggal sebagaimana yang dapat diterima Excel.

C4 fx =DATEVALUE(B4)-DATEVALUE(A4)

	A	B	C
1	MENGGUNAKAN DATEVALUE UNTUK		
2	MENGHITUNG HARI ANTARA DUA TANGGAL		
3	Tgl. Mulai (dlm format teks)	Tgl. Selesai (dlm format teks)	Jumlah Hari
4	11/3/2003	12/4/2003	31
5	01/21/04	03/13/2004	52
6	10/05/04	11/13/04	39
7	08/24/2004	09/04/2004	11
8	12/12/2003	03/13/2004	92
9	05/09/04	07/08/04	60
10	7/7/03	8/6/03	30
11	12/22/03	04/7/2004	107

Gambar 3-5: Worksheet yang dipergunakan untuk mengalkulasi jumlah hari di antara dua tanggal yang berformat teks.

2. Bloklah sel C4:C11, kemudian ketikkan formula sebagai berikut:
=DATEVALUE(B4)-DATEVALUE(A4)
3. Tekan <Ctrl+Enter>.

Catatan:

Jika hasil dari formula ini adalah #VALUE, maka kemungkinannya Anda memasukkan data tanggal dalam format, bukan teks. Oleh karenanya, baca kembali prosedur memasukkan data yang diuraikan di atas pada subbab ini.

3. Menggunakan Fungsi YEAR untuk Menampilkan Nilai Tahun dari Suatu Data Tanggal

Jika Anda memiliki sekumpulan data tanggal pada suatu sel, lalu Anda hanya menampilkan tahunnya saja pada suatu kolom, maka gunakanlah cara seperti yang dimuat pada worksheet Gambar 3.6.

Lakukan langkah-langkahnya sebagai berikut:

1. Pada kolom A4:A11 isikan data tanggal (sesuai dengan syarat masukan data tanggal ke Excel), data bisa dimasukkan dengan beberapa format.
2. Bloklah sel B4:B11 dan ketikkan formula =YEAR(A4).
3. Tekan <Ctrl+Enter>.

	A	B
1	MENGGUNAKAN FUNGSI YEAR UNTUK MENGEXTRACT DATA TAHUN SAJA	
2		
3	Data tanggal	Tahun
4	12/8/2010	2010
5	1-Mar-09	2009
6	8/1/2008	2008
7	Jan-09	2009
8	Nov-08	2008
9	Wednesday, June 11, 2008	2008
10	3-Feb	2010
11	9/12/2009	2009

Gambar 3-6: Meng-Extract data tahun dari suatu kolom.

4. Menggunakan Fungsi MONTH untuk Menampilkan Nama Bulan sebuah Data Tanggal

Menampilkan bulan dari suatu tanggal yang tersimpan dalam suatu kolom dengan memanfaatkan fungsi Month. Hasil dari fungsi tersebut berupa bilangan yang mewakili bulan dari Januari yang diwakili nilai 1 hingga Desember yang diwakili nilai 12. Untuk menampilkan ke format nama bulan secara lengkap, kita tidak dapat sekadar menggunakan format custom “MMMM”, melainkan dengan cara menggunakan fungsi CHOOSE dan MONTH. Kita coba mempraktikkan dari worksheet pada contoh kasus sebelumnya.

1. Gunakan worksheet pada contoh kasus sebelumnya.

2. Bloklah sel B4:B11, lalu ketikkan formula =MONTH(A4).
3. Tekan <Ctrl+Enter>. Rumus ini hanya akan menghasilkan bilangan yang mewakili nama bulan.

	A	B
3	Data tanggal	Tahun
4	12/8/2010	12
5	1-Mar-09	3
6	8/1/2008	8
7	Jan-09	1
8	Nov-08	11
9	Wednesday, June 11, 2008	6
10	3-Feb	2
11	9/12/2009	9

Gambar 3-7: Menampilkan bulan dalam bentuk bilangan dari suatu tanggal.

Contoh Penggunaan Fungsi Logika

1. Menggunakan Fungsi AND untuk Membandingkan Data di Dua Kolom

Sebuah data terdiri atas dua kolom. Kedua data pada kedua kolom ini perlu dievaluasi, apakah kolom pertama nilainya lebih besar dari 25 dan kolom kedua lebih besar dari 35. Jika kedua syarat tersebut benar, maka akan dihasilkan TRUE, sedangkan bila tidak, akan dihasilkan FALSE. Untuk pekerjaan ini digunakan fungsi AND yang cara penulisannya sebagai berikut:

=AND(logika1, logika2, ...)

- Logika1, logika2, ... argumennya 1 sampai 30 kondisi yang ingin Anda tes, apakah hasilnya TRUE atau FALSE.

- Argumennya harus berisi nilai logika atau array atau referensi yang berisi nilai logika.

Untuk mengerjakan latihan pada Gambar 3.8, langkahnya sebagai berikut:

1. Isikan pada kolom A dan B nilai secara acak antara 1 hingga 100.
2. Pada sel C3:C14 tuliskan formula sebagai berikut:
=AND (A3>25 , B3>35)
3. Tekan <Ctrl+Enter>.

	A	B	C
1			
2	Nilai 1	Nilai 2	Kondisi
3	50	37	TRUE
4	20	34	FALSE
5	14	27	FALSE
6	9	85	FALSE
7	5	90	FALSE
8	30	45	TRUE
9	70	39	TRUE
10	74	36	TRUE
11	58	3	FALSE
12	3	55	FALSE
13	15	63	FALSE
14	88	40	TRUE

Gambar 3-8: Memanfaatkan fungsi AND untuk membandingkan nilai dua kolom yang berisi data.

2. Menggunakan Fungsi OR untuk Memeriksa Sel yang Berisi Teks

Contoh kasus kali ini untuk memeriksa apakah data teks pada kolom A berisi “Excel 2007” atau Excel 2010”. Jika benar akan menghasilkan TRUE, dan jika salah akan menghasilkan FALSE.

Sekalipun data yang akan diperiksa berupa teks, kita juga bisa menggunakan fungsi AND. Namun kali ini kita akan memanfaatkan fungsi OR.

Fungsi ini dipergunakan untuk mengambil suatu keputusan di antara dua argumen yang diberikan. Fungsi ini akan menghasilkan TRUE (atau menjalankan pilihan) bila hasilnya benar. Sebaliknya, akan menghasilkan FALSE (atau tidak menjalankan pilihan) bila hasilnya salah.

Cara penulisannya:

=OR(logical1, logical2, ...)

- **Logika1, logika2, ...** argumennya 1 sampai 30 kondisi yang ingin Anda tes, apakah hasilnya TRUE atau FALSE.
- Argumennya harus berisi nilai logika atau array atau referensi yang berisi nilai logika.

	A	B	C	D
1	MENGUNAKAN FUNGSI OR			
2				
3	Nilai 1	Nilai 2	Hasil	
4	Excel 2003	Excel 2007	FALSE	
5	Excel 2010	Excel 2010	TRUE	
6	Excel 2003	Excel 2003	FALSE	
7	Excel 2007	Excel 2010	TRUE	
8	Excel 2007	Excel 2003	TRUE	
9	Excel 2010	Excel 2007	FALSE	
10	Excel 2010	Excel 2003	FALSE	
11	Excel 2007	Excel 2010	TRUE	
12	Excel 2003	Excel 2007	FALSE	
13	Excel 2007	Excel 2010	TRUE	
14	Excel 2010	Excel 2007	FALSE	
15	Excel 2003	Excel 2003	FALSE	

Gambar 3-9: Contoh penggunaan fungsi OR untuk mengevaluasi data Teks.

Pada contoh di atas, fungsi OR pada data teks membandingkan dua data dan keduanya harus benar agar memperoleh hasil TRUE. Jika salah satu tidak sesuai kriteria maka hasilnya FALSE.

Sekarang kita bandingkan bagaimana bila fungsi OR digunakan untuk menguji data yang berupa angka. Pada contoh berikut ini kita gunakan data pada kasus yang pertama dan fungsi penguji jika diganti dengan OR.

	A	B	C
17	MENGUNAKAN FUNGSI OR UNTUK ANGKA		
18			
19	Nilai 1	Nilai 2	Kondisi
20	50	37	TRUE
21	20	34	FALSE
22	14	27	FALSE
23	9	85	TRUE
24	5	90	TRUE
25	30	45	TRUE
26	70	39	TRUE
27	74	36	TRUE
28	58	3	TRUE
29	3	55	TRUE
30	15	63	TRUE
31	88	40	TRUE

Gambar 3-10: Menggunakan fungsi OR untuk memeriksa sel yang berisi angka.

Ternyata dari gambar di atas, kita melihat bahwa dengan fungsi OR akan menghasilkan nilai TRUE sekalipun data yang memenuhi hanya satu. Itulah maksud dari fungsi OR, logika 1 atau logika 2, sedangkan fungsi AND menguji Logika 1 dan Logika 2, kedua-duanya harus sesuai.

3. Menggunakan Fungsi IF untuk Menentukan Lulus atau Mengulang

Fungsi IF digunakan untuk membuat perhitungan dengan syarat. Misalnya pada daftar nilai ujian, kita bisa menentukan apakah siswa lulus atau harus mengulang. Dalam hal ini, nilai yang menjadi syarat

berupa nilai numerik 60. Dengan fungsi IF, Anda bisa menentukan bila lebih dari dan sama dengan (\geq) 60 maka artinya “Lulus”, sedangkan jika tidak, berarti “Mengulang”.

Penulisan formulanya sebagai berikut:

`=IF(B3>=60,"Lulus","Mengulang")`

C3			fx =IF(B3>=60,"Lulus","Mengulang")	
A	B	C	D	E
1				
2	Nama	Nilai	Hasil	
3	Mira	50	Mengulang	=IF(B3>=60,"Lulus","Mengulang")
4	Evi	75	Lulus	=IF(B3>=60,"Lulus","Mengulang")
5	Cesil	65	Lulus	=IF(B3>=60,"Lulus","Mengulang")
6	Desi	80	Lulus	=IF(B3>=60,"Lulus","Mengulang")
7	Amy	59	Mengulang	=IF(B3>=60,"Lulus","Mengulang")
8				

Gambar 3-11: Contoh penggunaan fungsi IF dengan syarat berupa bilangan.

4. Menggunakan Fungsi IF dengan Syarat Berupa Data Teks

Di atas kita telah menerapkan fungsi IF menggunakan syarat berupa data bilangan. Kali ini kita akan mencoba jika yang menjadi syarat berupa data teks.

Syarat kepegawaian menyatakan bila pendidikan SMA maka level kepegawaiannya sebagai “Karyawan”, sedang bila “S1” maka Staff.

Maka penulisan formulanya sebagai berikut:

`=IF(OR(C3="SMU",C3="S1"),"Staff","Karyawan")`

		D9		fx =IF(OR(C9="SMU",C9="S1"),"Staff","Karyawan")	
A	B	C	D	E	
1					
2	Nama	Pendidikan	Keterangan		
3	Parlan	S1	Staff	=IF(OR(C3="SMU",C3="S1"),"Staff","Karyawan")	
4	Pardi	SMA	Karyawan		
5	Parno	S1	Staff		
6	Parjo	S1	Staff		
7	Paiman	SMA	Karyawan		
8	Pono	S1	Staff		
9	Paidi	SMA	Karyawan		
10					
11	Syarat				
12	SMA	Karyawan			
13	S1	Staff			
14					

Gambar 3-12: Contoh penggunaan fungsi IF dengan syarat berupa data TEKS.

5. Menggunakan Fungsi IF untuk Menentukan Quartal dalam Satu Tahun

Berikut ini contoh jika syarat yang digunakan lebih dari satu. Untuk itu kita akan melakukan penggabungan fungsi IF dan OR secara bertingkat untuk menyebutkan secara otomatis suatu bulan itu termasuk kuartal yang ke berapa dalam suatu tahun.

Langkahnya sebagai berikut:

1. Pada sel A4 ketikkan January. Lalu pilih dan seretlah sel tersebut ke bawah untuk mendapatkan nama-nama bulan hingga A15 terisi December.
2. Pilihlah sel B4:B15, kemudian ketikkan dengan teliti formula yang panjang sebagai berikut:

```
=IF(OR(A4="January",A4="February",A4="March"), "Kwartal Pertama", IF(OR(A4="April",A4="May",A4="June"), "Kwartal Kedua", IF(OR(A4="July",A4="August",A4="September"), "Kwartal Ketiga", "Kwartal Keempat")))
```

3. Tekan <Ctrl+Enter>.

f_x		=IF(OR(A4="January",A4="February",A4="March"),"Kwartal	
	A	B	C
1	FUNGSI IF UNTUK Mencari Kwartal Tiap Tahun		
2			
3	BULAN	KWARTAL	
4	January	Kwartal Pertama	
5	February	Kwartal Pertama	
6	March	Kwartal Pertama	
7	April	Kwartal Kedua	
8	May	Kwartal Kedua	
9	June	Kwartal Kedua	
10	July	Kwartal Ketiga	
11	August	Kwartal Ketiga	
12	September	Kwartal Ketiga	
13	October	Kwartal Keempat	
14	November	Kwartal Keempat	
15	December	Kwartal Keempat	

Gambar 3-13: Pada worksheet di atas, formula yang kita buat secara otomatis mengenal nama-nama bulan dan menyebutkan kuartal yang ke berapa.

Catatan:

Sebagaimana aturan dasar penggunaan nested function (fungsi berganda), fungsi IF dapat dituliskan hingga 7 buah dalam suatu formula. Di dalam memberikan syarat pada logical_test, Anda dapat menggunakan semua operator perhitungan yang ada maupun operator perbandingan, seperti:

- < lebih kecil dari
- <= lebih kecil sama dengan
- > lebih besar dari
- >= lebih besar sama dengan
- = sama dengan
- <> tidak sama dengan

BAB 4

Mengelola Data Teks

Menyeragamkan Penulisan Data dengan Fungsi PROPER

Jika kita memiliki sebuah tabel data yang memuat daftar nama, taruhlah sejumlah 200 nama karyawan. Tetapi cara penulisan dilakukan secara tidak seragam, ada yang semuanya menggunakan huruf kapital, huruf campuran, huruf kecil dan sebagainya. Tugas kita membenahi cara penulisan nama-nama tersebut sesuai dengan kaidah yang benar.

	A	B
1	Sebelum Dibenahi	Setelah Dibenahi
2	HARRY K. SUNARYO	
3	JOHNI P BUANA	
4	Mirna Suryosaputro	
5	IR. MIEKE AMBARUKMI	
6	Angelina MARGARET	
7	SIDDHARTA Chandra	
8	tan hendra prasaja	
9		

Gambar 4-1: Data-data yang ditulis dengan berbagai cara yang tidak seragam.

Jika kita harus mengetik ulang data tersebut, maka akan membuang waktu lebih dari satu jam. Tetapi untungnya Excel memiliki fungsi teks yang memudahkan, yaitu fungsi PROPER.

Penyelesaiannya sebagai berikut:

1. Pada sel B2, ketikkan **=PROPER(A2)** tekan Enter.
2. Maka Anda akan memperoleh hasil Harry K. Sunaryo.
3. Copy-kan sel B2 ke B3:B201, maka pekerjaan Anda beres dalam waktu satu menit.



	A	B
1	Sebelum Dibenahi	Setelah Dibenahi
2	HARRY K. SUNARYO	Harry K. Sunaryo
3	JOHNI P BUANA	Johni P Buana
4	Mirna Suryosaputro	Mirna Suryosaputro
5	IR. MIEKE AMBARUKMI	Ir. Mieke Ambarukmi
6	Angelina MARGARET	Angelina Margaret
7	SIDDHARTA Chandra	Siddharta Chandra
8	tan hendra prasaja	Tan Hendra Prasaja
9		

Gambar 4-2: Berbagai macam cara penulisan data yang tidak seragam telah dibenahi dengan rapi sesuai kaidah penulisan yang benar.

Penjelasan Tentang Fungsi PROPER

Fungsi PROPER untuk mengubah sekumpulan kata (karakter) menjadi berawalan huruf kapital pada setiap awal kata dan mengubah teks berikutnya ke huruf kecil.


Cara Penyusunan Argumen Fungsi PROPER

=PROPER(teks)

Di mana argumen *teks* adalah teks atau alamat sel yang berisi teks yang akan dikonversikan.

Memisahkan Kodepos dari Alamat dengan Fungsi RIGHT, LEFT, dan LEN

Pada gambar tabel berikut ini menunjukkan daftar alamat yang telah disertai dengan kodeposnya. Tugas kita adalah memisahkan antara data alamat saja dengan nomor kodeposnya.



	A	B
1	Alamat	
2	Kosagrha Kel. Medokan Ayu Kec. Rungkut 60295	
3	Medayu Selatan Kel. Medokan Ayu Kec. Rungkut 60295	
4	Medayu Utara Kel. Medokan Ayu Kec. Rungkut 60295	
5	Medokan Ayu Kec. Rungkut Kec. Rungkut 60295	
6	Medokan Ayu Kel. Medokan Ayu Kec. Rungkut 60295	
7	Wonorejo Permai Kel. Wonorejo Kec. Rungkut 60296	
8	Wonorejo Eksekutif Kel. Wonorejo Kec. Rungkut 60296	
9	Wonorejo Kec. Rungkut Kec. Rungkut 60296	
10	Wonorungkut Kel. Wonorejo Kec. Rungkut 60296	
11	Pandugo Baru Blok A-H Kel. Penjaringansari Kec. Rungkut 60297	
12	Penjaringan Sari Kel. Penjaringansari Kec. Rungkut 60297	
13	Penjaringansari Kec. Rungkut Kec. Rungkut 60297	
14	Wisma Penjaringansari Kel. Penjaringansari Kec. Rungkut 60297	
15	Baruk Utara I - XV Kel. Kedungbaruk Kec. Rungkut 60298	
16	Beringin Kel. Kedungbaruk Kec. Rungkut 60298	
17		

Gambar 4-3: Alamat dan kodepos masih menyatu dalam satu sel.

Dalam hal ini ada dua tahap pekerjaan, yang pertama mengambil kodepos dari masing-masing alamat dan meletakkan pada sel tersendiri. Yang kedua adalah mengambil alamatnya saja dan meletakkan pada sel tersendiri.

Mengambil data Kodepos:

1. Pada tabel seperti gambar di atas, bloklah sel B2:B16.
2. Ketikkan formula: `=RIGHT(A2;5)`
3. Tekan `<Ctrl+ENTER>`.

Mengambil data alamatnya saja:

4. Bloklah sel C2:C16.
5. Ketikkan formula: =LEFT(A2;(LEN(A2)-LEN(B2)))
6. Tekan <Ctrl+ENTER>.

Hasilnya seperti pada gambar berikut.

C2			
fx =LEFT(A2;(LEN(A2)-LEN(B2)))			
	A	B	C
	Alamat & Kode Pos	Kode Pos	Alamat
2	Kosagrha Kel. Medokan Ayu Kec. Rungkut 60295	60295	Kosagrha Kel. Medokan Ayu Kec. Rungkut
3	Medayu Selatan Kel. Medokan Ayu Kec. Rungkut 60295	60295	Medayu Selatan Kel. Medokan Ayu Kec. Rungkut
4	Medayu Utara Kel. Medokan Ayu Kec. Rungkut 60295	60295	Medayu Utara Kel. Medokan Ayu Kec. Rungkut
5	Medokan Ayu Kec. Rungkut Kec. Rungkut 60295	60295	Medokan Ayu Kec. Rungkut Kec. Rungkut
6	Medokan Ayu Kel. Medokan Ayu Kec. Rungkut 60295	60295	Medokan Ayu Kel. Medokan Ayu Kec. Rungkut
7	Wonorejo Permai Kel. Wonorejo Kec. Rungkut 60296	60296	Wonorejo Permai Kel. Wonorejo Kec. Rungkut
8	Wonorejo Eksekutif Kel. Wonorejo Kec. Rungkut 60296	60296	Wonorejo Eksekutif Kel. Wonorejo Kec. Rungkut
9	Wonorejo Kec. Rungkut Kec. Rungkut 60296	60296	Wonorejo Kec. Rungkut Kec. Rungkut
10	Wonorungkut Kel. Wonorejo Kec. Rungkut 60296	60296	Wonorungkut Kel. Wonorejo Kec. Rungkut
11	Pandugo Baru Blok A-H Kel. Penjaringsari Kec. Rungkut 60297	60297	Pandugo Baru Blok A-H Kel. Penjaringsari Kec. Rungkut
12	Penjaringan Sari Kel. Penjaringsari Kec. Rungkut 60297	60297	Penjaringan Sari Kel. Penjaringsari Kec. Rungkut
13	Penjaringsari Kec. Rungkut Kec. Rungkut 60297	60297	Penjaringsari Kec. Rungkut Kec. Rungkut
14	Wisma Penjaringsari Kel. Penjaringsari Kec. Rungkut 60297	60297	Wisma Penjaringsari Kel. Penjaringsari Kec. Rungkut
15	Baruk Utara I - XV Kel. Kedungbaruk Kec. Rungkut 60298	60298	Baruk Utara I - XV Kel. Kedungbaruk Kec. Rungkut
16	Beringin Kel. Kedungbaruk Kec. Rungkut 60298	60298	Beringin Kel. Kedungbaruk Kec. Rungkut

Gambar 4-4: Hasil kedua formula yang kita buat di atas.

Penjelasan Tentang Fungsi RIGHT

Untuk menyalin karakter pada suatu sel dihitung mulai pada posisi paling kanan, sejumlah karakter yang dikehendaki.

Cara penyusunan argumen fungsi ini:

=RIGHT(Teks, Jumlah karakter)

Teks, adalah teks atau alamat sel yang berisi teks.

Jumlah karakter, adalah jumlah karakter yang akan disalin, dihitung dari posisi paling kanan.

Catatan: Apabila *Jumlah_karakter* melebihi jumlah karakter dalam *Teks*, maka keseluruhan *Teks* akan ditampilkan, sedang apabila argumen *Jumlah_karakter* tidak ditulis, akan dianggap sebagai 1 karakter.

Penjelasan Tentang Fungsi LEFT

Digunakan untuk memotong (mengambil) teks dari kiri sebanyak yang Anda tulis.

Cara Penyusunan Argumen Fungsi:

=LEFT(teks, Jumlah_karakter)

Teks, adalah data Teks yang berisi karakter yang akan diambil dengan fungsi ini. Data teks dapat digantikan dengan referensi suatu alamat sel.

Jumlah_karakter, menunjukkan berapa jumlah karakter dari kiri yang akan diambil.

Catatan: apabila Jumlah_karakter melebihi jumlah karakter dalam Teks, maka keseluruhan Teks akan ditampilkan. Apabila Jumlah_karakter tidak ditulis, maka akan dianggap sebagai 1.

Penjelasan Tentang Fungsi LEN

Untuk menghitung jumlah karakter yang terdapat dalam suatu data teks. Jumlah yang dihitung melalui fungsi ini termasuk spasi kosong bila ada.

Menggabungkan Dua atau Lebih Data Teks

Berbeda dengan data bilangan (value), pada data teks Excel menggunakan operator & untuk menggabungkan dua data. Sebagai contoh, jika sel A1 berisi teks Surabaya dan A2 berisi Jakarta, maka jika kita mengetikkan formula:

=A1&A2

Akan menghasilkan SurabayaJakarta.

Jika kita menghendaki jarak antardata, kita harus memasukkan tanda spasi pada formula sebagai berikut:

=A1&" "&A2, hasilnya **Surabaya Jakarta**.


Maupun membubuhkan koma di antara kedua data dengan cara penulisan sebagai berikut:

=A1&"", " &A2, hasilnya **Surabaya, Jakarta**.

Selain menggunakan tanda kutip untuk menyisipkan spasi, kita juga bisa memanfaatkan fungsi CHAR. Misalnya =CHAR(44) untuk menyisipkan koma, sedangkan =CHAR(32) untuk menyisipkan spasi. Dengan demikian penggunaannya sebagai berikut:

=A1&CHAR(44)&CHAR(32)&A2 hasilnya **Surabaya, Jakarta**.

Jika Anda ingin menyisipkan line break agar melipat ke baris berikutnya, gunakan =CHAR(10), sehingga formulanya menjadi: =A1&CHAR(10)&A2.



	A	B	C	D
1	Surabaya			
2	Jakarta			
3				
4				Surabaya Jakarta
5				

Gambar 4-5: Menyisipkan Line Break pada Excel 2007.


Tetapi ketika penulis menggunakan formula ini pada Excel 2010, hasilnya tidak berpindah ke baris berikutnya. Setelah klik tombol Wrap Text pada tab Home, baru melipat ke bawah.

Menggabungkan Data Teks, Tanggal dan Waktu dalam sebuah Data

Pada contoh ini kita akan menggabungkan data pada sel-sel yang masing-masing berisi teks, tanggal dan jam ke dalam sebuah sel.

Untuk memperoleh hasil yang diinginkan, Excel memiliki trik yang dapat Anda manfaatkan.

Berikut ini sel-sel yang berisi data yang akan digabungkan.




	A	B	C
1	Menggabungkan Data Teks, Tanggal dan Waktu		
2			
3			
4	Jadwal Pertandingan	PERSEBAYA VS AREMA	
5	Hari	Sabtu	
6	Tanggal		30-Apr-11
7	Jam		15:15
8			

Gambar 4-6: Worksheet yang berisi data-data yang akan dikombinasikan ke dalam sebuah sel.

Langkah-langkah untuk mengerjakan:

1. Pilih sel A8.
2. Ketikkan formula `=A3&" "&"Liga Jarum Black "&" "&" antara "&B3&" diadakan hari " &B4 &" tgl " &TEXT(B5;"dd-mmm-yyyy")&" tepat pada "&TEXT(B6;"hh:mm")&" !!"`
3. Tekan ENTER.

Hasilnya pada sel A8 akan diperoleh teks sebagai berikut:
"Jadwal Pertandingan Liga Jarum Black antara PERSEBAYA VS AREMA diadakan hari Sabtu tgl 30-Apr-2011 tepat pada 15:15!!"



Gambar 4-7: Hasil penggabungan antara teks dan data-data pada sel yang berisi teks, tanggal dan waktu.

Menggunakan Fungsi Search dan Replace untuk Membenahi Pengetikan Data yang Salah

Pada contoh berikut ini sebuah tabel yang berisi daftar jam kerja, antara jam mulai dan jam selesai. Tetapi data tersebut terketik dalam format waktu yang tidak dapat diterima oleh Excel. Kita juga tidak bisa menggunakan fungsi TIMEVALUE (dibahas pada bab tentang fungsi waktu) karena bukan sekadar data teks. Data yang cara pengetikannya sekalipun dapat kita mengerti, tetapi tidak bisa diterima sistem Excel. Misalnya untuk mengetikkan data jam dan menit yang seharusnya dipisahkan dengan titik dua, tetapi diketik dengan titik saja.

	A	B	C
1	Mulai	Selesai	Total
2	7.30	14.00	
3	05.45	14.30	
4	07.00	13.56	
5	08.15	17.45	
6	08.09	14.33	
7	7.59	12.15	
8	6.30	15.00	
9	07.01	14.20	
10	05.55	15.50	

Gambar 4-8: Tabel dengan data waktu yang diketik tidak sesuai ketentuan Excel sehingga tidak bisa dihitung dengan fungsi Excel.

Yang dapat kita lakukan adalah membenahi data tersebut agar menjadi format yang dimengerti Excel dengan langkah berikut:


1. Pada worksheet yang terlihat seperti gambar di atas, bloklah C2:C10.
2. Ketikkan formula sebagai berikut:

$$=(REPLACE(B2;SEARCH(" ";B2);1;"")-REPLACE(A2;SEARCH(" ";A2);1;""))$$
3. Tekan <Ctrl+Enter>.

	A	B	C
1	Mulai	Selesai	Total
2	7.30	14.00	0,270833333
3	05.45	14.30	0,364583333
4	07.00	13.56	0,288888889
5	08.15	17.45	0,395833333
6	08.09	14.33	0,266666667
7	7.59	12.15	0,177777778
8	6.30	15.00	0,354166667
9	07.01	14.20	0,304861111
10	05.55	15.50	0,413194444

Gambar 4-9: Hasil formula dalam format general.

- Ubahlah format bilangan hasil dari formula di atas dengan custom format sebagai berikut: *h:mm:ss*



Gambar 4-10: Menggunakan format waktu.

- Hasil yang diperoleh akan nampak pada gambar sebagai berikut.

	A	B	C
	Mulai	Selesai	Total
1			
2	7.30	14.00	6:30:00
3	05.45	14.30	8:45:00
4	07.00	13.56	6:56:00
5	08.15	17.45	9:30:00
6	08.09	14.33	6:24:00
7	7.59	12.15	4:16:00
8	6.30	15.00	8:30:00
9	07.01	14.20	7:19:00
10	05.55	15.50	9:55:00

Gambar 4-11: Setelah kolom C diformat dengan format waktu.

Menghilangkan Sebutan Bp/Bpk/Ibu/Sdr di depan Nama lalu Memisahkan Nama Depan dan Belakang

Sebuah daftar berisi nama-nama orang di mana di depannya diawali dengan sebutan Bp, Bpk, Ibu maupun Sdr. Tugas kita adalah menghapus sebutan tersebut pada data sehingga setelah diedit, data yang baru tinggal nama saja tanpa sebutan di depannya. Setelah itu memisahkan nama depan dari belakang.

	A	B
1	Menghilangkan sebutan Bp dan Ibu pada data	
2		
3		
4	Sebelum di Edit	Setelah di Edit
5	Ibu Susan Murdiono	
6	Bp Hadi Wirapati	
7	Bpk Aad Surapraja	
8	Ibu Gunawan Mochtar	
9	Sdr Yan Arifin	
10	Bpk Leonardo Widya	
11	Ibu Mince Harimurti	
12	Sdr. Ade kusuma	
13		

Gambar 4-12: Data awal yang akan dibenahi.

Langkah Menghilangkan Sebutan

1. Berdasarkan tabel pada Gambar 4-15, bloklah sel **B5:B12**.
2. Ketikkan formula sebagai berikut:
$$=IF(OR(LEFT(A5;2)="Bp";LEFT(A5;3)="Ibu";LEFT(A5;3)="Sdr");RIGHT(A5;LEN(A5)-FIND(" ";A5));A5)$$
3. Tekan <Ctrl+Enter>.

fx =IF(OR(LEFT(A5;2)="Bp";LEFT(A5;3)="Ibu";LEFT(A5;3)="Sdr");RIGHT(A5;LEN(A5)-FIND(" ";A5));A5)				
A	B	C	D	E
1	Menghilangkan sebutan Bp dan Ibu pada data			
2				
3				
4	Sebelum di Edit	Setelah di Edit		
5	Ibu Susan Murdiono	Susan Murdiono		
6	Bp Hadi Wirapati	Hadi Wirapati		
7	Bpk Aad Surapraja	Aad Surapraja		
8	Ibu Gunawan Mochtar	Gunawan Mochtar		
9	Sdr Yan Arifin	Yan Arifin		
10	Bpk Leonardo Widya	Leonardo Widya		
11	Ibu Mince Harimurti	Mince Harimurti		
12	Sdr. Ade kusuma	Ade kusuma		
13				

Gambar 4-13: Kolom B memperlihatkan data setelah dihilangkan sebutan di depan nama.

Langkah Memisahkan Nama Depan

1. Bloklah sel C5:C12.
2. Ketikkan formula sebagai berikut: =LEFT(B5;SEARCH(" ";B5)-1)
3. Tekan <Ctrl+Enter>.

C5 fx =LEFT(B5;SEARCH(" ";B5)-1)		
A	B	C
4	Sebelum di Edit	Setelah di Edit
5	Ibu Susan Murdiono	Susan Murdiono
6	Bp Hadi Wirapati	Hadi Wirapati
7	Bpk Aad Surapraja	Aad Surapraja
8	Ibu Gunawan Mochtar	Gunawan Mochtar
9	Sdr Yan Arifin	Yan Arifin
10	Bpk Leonardo Widya	Leonardo Widya
11	Ibu Mince Harimurti	Mince Harimurti
12	Sdr. Ade kusuma	Ade kusuma
13		

Gambar 4-14: Kolom C berisi Nama depan yang telah dipisahkan.

Langkah Memisahkan Nama Belakang

1. Bloklah sel D5:D12.
2. Ketikkan formula sebagai berikut:
$$=MID(B5;SEARCH(" ";B5)+1;100)$$
3. Tekan <Ctrl+Enter>.

	A	B	C	D
4	Sebelum di Edit	Setelah di Edit	Nama Depan	Nama Belakang
5	Ibu Susan Murdiono	Susan Murdiono	Susan	Murdiono
6	Bp Hadi Wirapati	Hadi Wirapati	Hadi	Wirapati
7	Bpk Aad Surapraja	Aad Surapraja	Aad	Surapraja
8	Ibu Gunawan Mochtar	Gunawan Mochtar	Gunawan	Mochtar
9	Sdr Yan Arifin	Yan Arifin	Yan	Arifin
10	Bpk Leonardo Widya	Leonardo Widya	Leonardo	Widya
11	Ibu Mince Harimurti	Mince Harimurti	Mince	Harimurti
12	Sdr. Ade kusuma	Ade kusuma	Ade	kusuma

Gambar 4-15: Kolom D menghasilkan Nama Belakang yang telah dipisahkan.

Memisahkan Nama yang Terdiri Atas 3 Bagian

Jika kita memiliki daftar nama-nama yang terdiri atas tiga bagian, yaitu nama depan, nama tengah dan nama belakang. Misalnya Ang Tjeng Yan, Dewi Siti Aminah, John Mc Cullay, maka kita memiliki kiat memisahkan yang berbeda.

Memisahkan nama depan pada sel B4

Ketikkan formula sebagai berikut:

$$=LEFT(A6;FIND(" ";A6)-1)$$

Formula ini menggunakan fungsi LEFT untuk mengambil nama depan. Caranya menghitung jumlah karakter dari kiri sampai menemukan tanda spasi. Fungsi FIND yang diletakkan sebagai argumen kedua dari fungsi LEFT adalah mencari di manakah posisi

spasi kosong. Setelah menemukan spasi kosong terletak di posisi ke berapa, kemudian dikurangi satu. Maksudnya hanya menghitung posisi karakter terakhir pada nama depan.

Jadi, untuk mengambil kata **Ang**, fungsi LEFT akan membuat perhitungan dengan fungsi FIND yang menemukan posisi spasi kosong berada pada posisi ke-4, lalu dikurangi 1.

Memisahkan Nama Tengah pada Sel C4

```
=IF(ISERR(MID(A6;FIND(" ";A6)+1;IF(ISERR(FIND(" ";A6;FIND(" ";A6)+1));FIND(" ";A6);FIND(" ";A6;FIND(" ";A6)+1))-FIND(" ";A6)-1));";MID(A6;FIND(" ";A6)+1;IF(ISERR(FIND(" ";A6;-FIND(" ";A6)+1));FIND(" ";A6);FIND(" ";A6;FIND(" ";A6)+1))-FIND(" ";A6)-1))
```

Formula di atas menggunakan fungsi IF untuk mencari posisi spasi kosong (" ") batas nama depan dan tengah plus satu. Maksudnya untuk menandai karakter pertama nama tengah, posisinya pada karakter yang ke berapa. Dari posisi itu mencari spasi kosong yang kedua. Setelah ketemu posisinya lalu dikurangkan dengan posisi spasi kosong yang pertama sehingga memperoleh nilai jumlah karakter nama tengah. Dengan fungsi MID, rangkaian fungsi di atas menghasilkan **Tjen**.

Memisahkan nama belakang pada sel D4

```
=RIGHT(A6;LEN(A6)-FIND(" *";SUBSTITUTE(A6;" *";LEN(A6)LEN(SUBSTITUTE(A6;" ";")))))
```

	A	B	C	D
1	Memisahkan Nama Depan, Tengah dan Belakang			
2				
3	Nama Lengkap	Nama Depan	Nama Tengah	Nama Belakang
4	Ang Tjen Yan	Ang	Tjen	Yan
5	Dewi Siti Aminah	Dewi	Siti	Aminah
6	John Mac Cullay	John	Mac	Cullay
7				

Gambar 4-16: Daftar Nama yang terdiri atas tiga bagian, depan, tengah dan belakang.

Keterangan Tentang Fungsi ISERR

Fungsi ini gunanya untuk menguji apakah tipe nilai atau referensi yang ada pada suatu sel atau range. Dari fungsi ini dihasilkan TRUE atau FALSE. Fungsi ISERR untuk menguji jika pada Nilai ditemukan nilai kesalahan (kecuali #N/A!).

Cara penulisan argumen fungsi ini:

=ISERR(nilai yang diuji)

Keterangan Tentang Fungsi FIND

Untuk mencari suatu karakter dalam suatu kumpulan teks string dan akan menghasilkan jumlah karakter dari teks yang dicari. Fungsi ini mirip dengan fungsi SEARCH, hanya bedanya FIND akan membedakan huruf besar (kapital) dan huruf kecil, dan tidak dapat menerima pencarian dengan karakter wildcard.

Cara penulisan argumen fungsi ini:

=FIND(teks yang dicari, dalam sekumpulan teks, dimulai dari karakter yang beberapa)

Teks yang dicari adalah teks yang akan dicari dengan bantuan fungsi ini. Jika yang dicari adalah teks kosong (""), maka fungsi ini akan mencocokkan karakter pertama dari teks yang dicari. Tidak boleh diisi dengan karakter wild card.

Dalam sekumpulan teks adalah sekumpulan teks yang mengandung teks yang dicari.

Dimulai dari karakter yang ke berapa pencarian dimulai dari karakter yang ke berapa dari sekumpulan teks. Jika argumen ini tidak ditulis maka akan dianggap pencarian dimulai dari karakter pertama.

Contoh lain dalam penggunaan fungsi FIND:

=FIND("m", "Muhamad Yusron") menghasilkan 5

=FIND("Y";"Muhamad Yusron";8) menghasilkan 9